

C1M0

Introduction to Maple

Our discussion will focus on Maple 6, which was developed by Waterloo Maple Inc. in Waterloo, Ontario, Canada. Quoting from the *Maple 6 Learning Guide*, “Maple is a *Symbolic Computation System* or *Computer Algebra System*. Both phrases refer to Maple’s ability to manipulate information in a symbolic or algebraic manner. Conventional mathematical programs require numerical values for all variables. By contrast, Maple maintains and manipulates the underlying symbols and expressions, as well as evaluates numerical expressions.” From the second description you see why Maple is designated as a ‘CAS’.

Assignment Format

We are going to begin by establishing a format for each Maple assignment that is to be handed in. Open Maple and obtain a blank worksheet. Do not type the “<” or “>” which are shown to identify your entries. And <Enter> means the “Enter” key. As you begin, the worksheet is in “math mode”, so ‘click’ on the **T** to switch into “text mode”.



For the assignment **C1M1**, type <C1M1> <Enter> and then highlight C1M1 and click on the middle of the three boxes to the right of **B** *I* u so as to center C1M1. The left of these three buttons left-justifies text and the right one right-justifies it. Now,

<down arrow>, then type your name and section as shown.

<Midn Your Name> <Enter>

<Section> <Your section> <Enter>

Having completed this, highlight the three lines and then click on **B** to boldface everything. This is the format you should use for all Maple assignments to be handed in. For example, you should see something like

C1M1

Midn John Doe

Section 1234

Beginning Maple Syntax

Since we are building the foundation for the use of Maple, we will designate this spadework by ♠ when we wish to highlight an important fact. Since getting on-line help is extremely important, we will begin with that. Suppose that you have a question about the command ‘plot’. Then in a worksheet enter <?plot> <Enter> and you will see the information available and links for other related topics.

♠ To obtain on-line help on ‘command’, enter <?command> <Enter>.

You may eliminate the brackets on the left by pressing the function key <F9>. To return to math mode, click on the **[>]**. If we wanted to type a math formula while in text mode we would click on **[Σ]**. Later in this section we will discuss *palettes* which allow you to select commands from a menu and avoid using Maple syntax. It is the contention of the author of these notes that learning some Maple syntax is beneficial to the student, so even though you may accomplish the same things by clicking on a symbol, we will show you the syntax that would otherwise be hidden.

In math mode, lines in Maple end with a semicolon or colon.

♠ If a line ends with a semicolon, then the output will be displayed.

♠ If a line ends with a colon, then the display of the output is suppressed.

♠ To activate a line, press <Enter> with the cursor at **any** position on that line. (This does not break the line as it would in a word processor.)

Please type the command lines below in a new worksheet exactly as you see them, remembering to press <Enter>, and note the output. This work is for your benefit and is not intended to be handed in.

```
> a:=4;
> sqrt(a);
> b=4;
> sqrt(b);
```

We did not display the output here because it is important that the reader discover the results for themselves. However, the concept is very important. The first line shows the format for assigning a value to the variable ‘ a ’. Think of this as placing the value 4 in a memory location, named ‘ a ’, which can be retrieved when needed. The third line is an equation. While it is trivial to solve for b , the value of b is not accessible in this format.

♠ Use `a:=b` to assign the value ‘ b ’ to ‘ a ’. Note that ‘ b ’ may be a number or an expression.

Expression versus Function

Please enter the following lines in a worksheet, remembering to press <Enter> to activate each line:

```
> A:=x^2+sin(4*x);
> f:=x->sqrt(x^2+9);
> subs(x=4,A);
> f(x); f(4);
> subs(x=4,f(x));
> simplify(%);
```

The first line names the **expression** $x^2 + \sin(4x)$ as ‘ A ’. Note how ‘ $*$ ’ must be used when two factors are to be multiplied. Omitting the $*$ is a common error for beginning Maple users. The second line shows how to define a **function** f . The third line shows how to substitute 4 for x in the expression A . Note: You may **not** use $4 = x$ and expect the same result. This part of the substitution must be of the form ‘old’ = ‘new’, because the ‘old’ will be replaced by the ‘new’. Try reversing the order of the substitution by entering `<subs(4=x,A)>`. What difference do you see?

The fourth line above is more important than it appears to be. You see that $f(x)$ is an **expression** and then the fifth line reinforces that. However, were you surprised at the output of the fifth line? There are many levels of operations and simplifications in Maple and substitution does not simplify as far as is possible. The use of % as the **simplify** argument refers to the previous output, which was $\sqrt{25}$.

♠ You must be aware of when the syntax calls for an expression or for a function.

♠ When two factors are to be multiplied, a ‘ $*$ ’ must separate them, i.e. `a*b`.

♠ You may use % to refer to the previous output in a Maple command. Warning! See comments about order of execution later in this section.

In that same worksheet enter these command lines and observe the output.

```
> value(Pi^2/6 +sin(Pi/6));
                                      $\frac{1}{6}\pi^2 + \frac{1}{2}$ 
> evalf(Pi^2/6+sin(Pi/6));
                                     2.144934068
> evalf(Pi^2/6+sin(Pi/6),75);
2.14493406684822643647241516664602518921894990120679843773555822937000747041
```

The ‘**value**’ command returns an exact value for an expression, while the **evalf** command converts an exact numerical expression to a floating point number. You also have the option of specifying the number of digits displayed, as shown on the third line. The default is to display ten digits. You may also set the number of digits to be displayed in the remainder of a worksheet by using a command, for example:

```
> Digits:=20;
                                     Digits := 20
> evalf(Pi);
                                     3.1415926535897932385
> evalf(pi);
                                      $\pi$ 
```

♠ In Maple, $\pi = \text{Pi}$ is a number, and `pi` = π is a small Greek letter with no numerical value.

It will be very useful later to be able to make a function out of an expression. The syntax for this is puzzling. Entering these commands should produce these results:

```
> P:=x^2+cos(x);
```

$$P := x^2 + \cos(x)$$

```
> G:=unapply(P,x);
```

$$G := x \rightarrow x^2 + \cos(x)$$

```
> G(Pi);
```

$$\pi^2 - 1$$

We see that G is a function and that $G(x) = P$.


Maple Graphics

Maple graphics are versatile and easy to use. Let's define $F(x) = 4 - 2x - x^2$ in our worksheet and see how we can get a quick plot of F on $[-2, 3]$. To save space we have included two plots side-by-side. The output of the first is on the left.


```
> F:=x->4-2*x-x^2;
```

$$F := x \rightarrow 4 - 2x - x^2$$

```
> plot(F(x),x=-2..3,color=blue);
```



Plot of $F(x)$



Plot of P

Note how we used $F(x)$ which is an expression, not just F , in the first plot. Now click on the displayed graph. A box will appear around the graph with small black boxes placed strategically. Move the cursor over the lower righthand box and position it so that a diagonal arrow appears. Click when the arrow is displayed and drag the arrow towards the center of the box, thereby resizing the box. You can change the aspect of the graph by making the box tall or short. You should always make the plots on your homework smaller so as to save paper.


♠ You can resize a Maple plot by clicking on the plot and then dragging the corner of the displayed box.

To plot the function G from above we could use P or $G(x)$ and obtain identical results. The output is on the right above.


```
> plot(P,x=0..Pi,color=magenta);
```

When you wish to plot two functions with the same domain it can be done very easily. However, it is also very easy to confuse this syntax with that of parametric plotting. We will do an example of each so that you will know where to be careful. The placement of the righthand square bracket determines which format you have. In two-dimensional plotting, when you list two expressions and a range inside the square brackets the first function controls the value on the x -axis and the second function controls the value on the y -axis. This is parametric plotting. To save space the output follows on the left.

```
> plot([exp(x),x^2,x=-1..1]);
```



Parametric Plotting



Two Functions

When you do not include the domain inside the square brackets you get two different plots on the same coordinate system as you can see above on the right. This was produced by:

```
> plot([exp(x),x^2],x=-1..1);
```

♠ To plot the graphs of two functions with the same domain on the same axes, include both functions in square brackets, but exclude the range. If the range is included in the brackets, then the result will be a parametric plot.

♠ The exponential function, e^x , is accessed by `exp(x)` in Maple, which does not recognize e as any particular number. See the following:

```
> evalf(e);
```

e

```
> evalf(exp(1),90);
```

```
2.71828182845904523536028747135266249775724709369995957496696762772407663035354759457138217
```

Maple Packages

Just as specialized mechanics require special tools, and they do not carry every trade's tools with them at all times, Maple has packaged different commands into different libraries so that not all commands need to be put into active memory at all times. Instead, the user may select one or more packages as needed and thereby save computer memory space. For a list of all the packages, see section 3.8 on page 104 of the *Maple 6 Learning Guide*. Or, in a worksheet type `<?packages>` to see the listing in the *Help* section of Maple.

There are several packages that the beginning calculus student will need. The first is `student`, which contains many calculus operations, and the second is `plots`. As you would expect, `plots` is a graphics package. To invoke, or 'call up' a package you would insert a line in the worksheet before the package is needed such as:

```
> with(plots):
```

If you had ended the line above with a semi-colon, then a list of all the commands in `plots` would have been displayed. You should probably try this once just to see what is there. Also, try putting a semi-colon after `with(student)`.


More Graphics

How should you plot two functions which have different domains on the same coordinate axes? The answer is to give each plot a name **using a colon at the end to suppress the output** and then display them together. The command `display` is in the graphics package `plots`.

```
> with(plots):
```

```
> A:=plot(exp(x),x=-3..3): ← colon!
```

```
> B:=plot(ln(x),x=.01..4): ← colon!
> display(A,B); ← semicolon!
```




Order of Execution of Commands


Sometimes previous experience with computers causes us to assume that the order of execution of the commands in Maple is the same as the order shown on the screen. Also, we assume that whatever we see in front of us has been executed. Both assumptions are false in Maple. Suppose that you have a worksheet that you had saved previously and that you have just reopened it. You move the cursor down to some line in the worksheet and hit <Enter>. As far as Maple is concerned, this is the first line executed and if it depends on lines above it, then it cannot execute correctly. Computationally, Maple remembers the information in the order in which the lines were executed.

Also, sometimes we want Maple to forget that it has done something and to start over. For this reason the first command line should have `restart:` as the first word. Then, call up the packages that you anticipate using. For example:

```
> restart: with(student): with(plots):
```

is a typical first command line in Calculus I. Since `restart:` clears the memory, what would happen if it occurred last on the line? That's right, the packages would be loaded and then erased from accessible memory.

It is quite reasonable and normal to move around in a worksheet, changing things as needed. In fact, one frequently realizes that a new command line needs to be inserted earlier in a worksheet. To do this, place the cursor on the line above where you want a new line. Then, click on the  button and a new blank command line will appear. But, when you execute this line by hitting <Enter>, what will be the order of execution of the Maple commands? How do you ensure that the order of execution of the commands is the same as what you see? One approach would be to put the cursor on the first line and hit <Enter> until you reach the last line. That is effective, but it is not necessary. Move the cursor up to 'Edit' on the left at the top of the screen, 'click', and then slide the cursor down to 'Execute'. A side panel will open, slide the cursor over and down to 'Worksheet', and click'. At this point, Maple will execute the worksheet in the order of the commands shown on the screen.

♠ To insert a new command line in a worksheet, put the cursor on the line above and click on .

♠ **Always** execute a worksheet upon reopening it and before additional commands are added.

♠ **Always** execute a worksheet before saving it and printing it out to be handed in.

Example: Suppose that we are given two points in the plane, $P_1(2,5)$ and $P_2(-1,1)$, and we wish to find the distance between them and an equation for the line that contains them.

```
> restart:
> x1:=2; y1:=5; x2:=-1; y2:=1;
```

```

x1 := 2
y1 := 5
x2 := -1
y2 := 1
> distance:=sqrt((x2-x1)^2+(y2-y1)^2);
distance := 5
> slope:=(y2-y1)/(x2-x1);
slope := 4/3
> line1:=y-y1=slope*(x-x1);
line1 := y - 5 = 4/3 x - 8/3
> y=solve(line1,y);
y = 7/3 + 4/3 x

```

The process is straightforward until we reach the line that begins with `line1`. We assigned the name 'line1' to the equation for the line using the format $y - y_0 = m(x - x_0)$, where m is the slope of the line and $P(x_0, y_0)$ is a known point on the line. We chose to use P_1 , but the choice of P_2 would have worked equally well. Then, we established a new equation with y being set equal to the solution of the equation `line1` for the variable y . This yields the format $y = mx + b$, which some prefer.

There are two operations that are very basic in calculus, namely differentiation and integration, or anti-differentiation. An expression in x and t , say $\tan(x/t)$, can be differentiated with respect to either variable, so we must remember to specify the variable with respect to which the operation is being performed. Using the expression $P = x^2 + \cos(x)$ from above we have

```

> Pprime:=diff(P,x);
Pprime := 2x - sin(x)

```

And if we integrate P

```

> Pint:=int(P,x);
Pint := x^3/3 + sin(x)

```

Now let's do some of the same steps by using a palette. On your command line type

`A:=`

There are three palettes and to access them you begin by clicking on "View", then "Palettes". If you need symbols, select that palette, but for now we choose "expressions". You should see



Click on the box with the integral symbol $\int a$. Then click on the box with a^b . On your command line the cursor appears where you want x inserted, so you type `< x >`, and then **move to the next entry position by using the 'Tab' key**. Enter `< 2 >`, `<Tab>`, and then the variable of integration, `< x >` and `<Enter>`. At the end of the command line put a semicolon and hit `<Enter>`. This should produce

```

> A:=int(x^2,x);
x^3

```

Exercises Use Maple to find the plots and answers.

1. Display the graph of $y = 3 \sin(2x + \pi/6) + 2$ on the interval $[-\pi, \frac{3\pi}{2}]$.
2. Define $f(x) = x^2 - 4x + 4$ and $g(x) = \ln(x + 2)$ as functions and in the same plot display their graphs for the interval $[-1, 4]$.
3. Find the value of e^{-2x} for $x = .7$ and display 25 places.